

## Sprachbeschreibung: Java<sub>SST</sub>

class = "class" ident classbody.

classbody = "{" declarations "}".

declarations = {"final" type ident "=" expression ";"}  
                  { type ident ";"}  
                  { method\_declaration }.

method\_declaration = method\_head method\_body.

method\_head = "public" method\_type ident formal\_parameters.

method\_type = "void" | type.

formal\_parameters = "(" [fp\_section {"," fp\_section}] ")"

fp\_section = type ident .

method\_body = "{" {local\_declaration} statement\_sequence"}".

local\_declaration = type ident ";"

statement\_sequence = statement {statement}.

statement = assignment | procedure\_call | if\_statement | while\_statement | return\_statement.

type = "int".

assignment = ident "=" expression ";"

procedure\_call = intern\_procedure\_call ";"

intern\_procedure\_call = ident actual\_parameters.

if\_statement = "if" "(" expression ")" "{" statement\_sequence "}" "else" "{" statement\_sequence "}"

while\_statement = "while" "(" expression ")" "{" statement\_sequence "}"

return\_statement = "return" [ simple\_expression ] ";"

actual\_parameters = "(" [expression {"," expression}] ")"

expression = simple\_expression [( "==" | "<" | "<=" | ">" | ">=" ) simple\_expression].

simple\_expression = term {( "+" | "-" ) term}.

term = factor {( "\*" | "/" ) factor}.

factor = ident | number | "(" expression ")" | intern\_procedure\_call.

number = digit {digit}.

ident = letter {letter | digit}.

digit = („0“ | ... | „9“).

letter = („a“ | ... | „z“ | „A“ | ... | „Z“).